eAppendix: WinBUGS code for Models 2 and 3 (online)

model 2 {

   for (i in 1:Nareas) {

      Y1[i] ~ dpois(mu[i, 1])                                # likelihood for copd
       log(mu[i, 1]) <- log(E1[i]) + eta[1, i] 

       Y2[i] ~ dpois(mu[i, 2])                               # likelihood for lung cancer
       log(mu[i, 2]) <- log(E2[i]) + eta[2, i] 


       # Centred version of BYM prior for log relative risks

       eta[1, i] ~ dnorm(theta[1,i], tau.psi[1])       # log relative risk for copd
       theta[1,i] <- alpha[1] + beta*phi[i] + psi1[i]    

       eta[2, i] ~ dnorm(theta[2,i], tau.psi[2])       # log relative risk for lung cancer
       theta[2,i] <- alpha[2] + phi[i]       

      # (Note: order of k and i index for psi and eta are switched as car.normal distribution

      #  and sd function can only be applied to columns of an array)

   }

   # CAR prior for the copd specific spatial random effects

   psi1[1:Nareas] ~ car.normal(adj[], weights[], num[], omega.psi)    

    # CAR prior for the lung cancer (shared) spatial random effects

   phi[1:Nareas] ~ car.normal(adj[], weights[], num[], omega.phi)  
   for (j in 1:sumNumNeigh) {    weights[j] <- 1 
}   # define adjacency weights for spatial CAR priors

   # Other priors

   for (k in 1:Ndiseases) {

      alpha[k] ~ dflat() 

            # intercept disease k
      tau.psi[k] ~ dgamma(0.5, 0.0005)     # precision of unstructured disease specific effects   

   }

   omega.psi ~ dgamma(0.5, 0.0005)     # precision of spatial copd specific effects    

   omega.phi ~ dgamma(0.5, 0.0005)     # precision of spatial lung cancer (shared) effects                 

   beta ~ dnorm(0, 0.00001)   # log RR of copd per unit change in log RR of lung cancer
   RR.smoke <- exp(beta)

   # Relative risks and other summary quantities 

   for (i in 1:Nareas) {

      copdtotalRR[i] <- exp(eta[1, i])     # overall RR of copd in area i

      lctotalRR[i] <- exp(eta[2, i])
       # overall RR of lung cancer in area i

      copdspecificRR[i]<- exp(psi1[i])   # residual RR of copd not explained by smoking proxy  

      sharedRR[i]<- exp(beta*phi[i])      # risk shared by copd and lung cancer

      pr.gt.1.copdspecificRR[i] <- step(copdspecificRR[i] - 1)   # Prob that copdspecificRR > 1

      pr.gt.1.sharedRR[i] <- step(sharedRR[i] - 1)                     # Prob that sharedRR > 1

   }
         

   var.shared <- pow(beta,2)*sd(phi[])*sd(phi[])
 # empirical variance of shared effects 

   var.specific.copd <- sd(psi1[])*sd(psi1[])
 # empirical variance of copd specific effects
   # fraction of total variation in RRs for each disease that is explained by the shared component

   frac.shared.copd <- var.shared / (var.shared + var.specific.copd)

   # total variance

   var.copd <- sd(eta[1,])*sd(eta[1,])

   var.lc <- sd(eta[2,])*sd(eta[2,])

   # overdispersion variance

   var.od[1] <- 1/tau.psi[1]

   var.od[2] <- 1/tau.psi[2]



}  

model 3 {

   for (i in 1:Nareas) {

      Y1[i] ~ dpois(mu[i, 1])                                # likelihood for copd
       log(mu[i, 1]) <- log(E1[i]) + eta[1, i] 

       Y2[i] ~ dpois(mu[i, 2])                               # likelihood for lung cancer
       log(mu[i, 2]) <- log(E2[i]) + eta[2, i] 


       # Centred version of BYM prior for log relative risks

       eta[1, i] ~ dnorm(theta[1,i], tau.psi[1])       # log relative risk for copd
       theta[1,i] <- alpha[1] + delta*phi[i] + psi[1, i]    

       eta[2, i] ~ dnorm(theta[2,i], tau.psi[2])       # log relative risk for lung cancer
       theta[2,i] <- alpha[2] + phi[i] /delta + psi[2, i]       

      # (Note: order of k and i index for psi and eta are switched as car.normal distribution

      #  and sd function can only be applied to columns of an array)

   }

   # CAR prior for the copd specific spatial random effects
   for (k in 1:Ndiseases) {   psi[k, 1:Nareas] ~ car.normal(adj[], weights[], num[], omega.psi[k])    }
    # CAR prior for the lung cancer (shared) spatial random effects

   phi[1:Nareas] ~ car.normal(adj[], weights[], num[], omega.phi)  
   for (j in 1:sumNumNeigh) {    weights[j] <- 1 
}   # define adjacency weights for spatial CAR priors
   # Other priors

   for (k in 1:Ndiseases) {

      alpha[k] ~ dflat() 

               # intercept disease k
      tau.psi[k] ~ dgamma(0.5, 0.0005)        # precision of unstructured disease specific effects   

      omega.psi[k] ~ dgamma(0.5, 0.0005)  # precision of spatial disease specific effects    

   }

   omega.phi ~ dgamma(0.5, 0.0005)         # precision of spatial shared effects                 

   logdelta ~ dnorm(0, 5.9)   # scaling factor for relative strength of shared component for each disease

   delta <- exp(logdelta) ;     RR.smoke <- exp(delta*delta)

   # Relative risks and other summary quantities 

   for (i in 1:Nareas) {

      copdtotalRR[i] <- exp(eta[1, i])     # overall RR of copd in area i

      lctotalRR[i] <- exp(eta[2, i])
       # overall RR of lung cancer in area i

      copdspecificRR[i]<- exp(psi[1,i])   # residual RR of copd not explained by shared component  
      lcspecificRR[i]<- exp(psi[2,i])        # residual RR of lung cancer not explained by shared component  

      sharedRR[i]<- exp(delta*phi[i])     # risk shared by both diseases (scaled for copd)
      pr.gt.1.copdspecificRR[i] <- step(copdspecificRR[i] - 1)   # Prob that copdspecificRR > 1

      pr.gt.1.sharedRR[i] <- step(sharedRR[i] - 1)                     # Prob that sharedRR > 1

   }
         

   var.shared.copd <- pow(delta,2)*sd(phi[])*sd(phi[]) # empirical var of shared effects (scaled for copd) 

   var.shared.lc <- pow(delta,-2)*sd(phi[])*sd(phi[])    # empirical var of shared effects (scaled for lc) 

   var.specific.copd <- sd(psi[1,])*sd(psi[1,])
     # empirical variance of copd specific effects
   var.specific.lc <- sd(psi[2,])*sd(psi[2,])
                  # empirical variance of lung cancer specific effects
   # fraction of total variation in RRs for each disease that is explained by the shared component

   frac.shared.copd <- var.shared.copd / (var.shared.copd + var.specific.copd)

   frac.shared.lc <- var.shared.lc / (var.shared.lc + var.specific.lc)
   # total variance

   var.copd <- sd(eta[1,])*sd(eta[1,])

   var.lc <- sd(eta[2,])*sd(eta[2,])

   # overdispersion variance

   var.od[1] <- 1/tau.psi[1]

   var.od[2] <- 1/tau.psi[2]



}  
